*&---------------------------------------------------------------------*
*& Subroutine pool   ZBW_Z_DS_PF
*& Author:           Raja Cherukuri
*&--------------------------------------------------------------------*

* Type Definition-----------------------------------------------------*

program  zbw_z_ds_pf .

data: t_dp type standard table of zoxcr20022.

*&---------------------------------------------------------------------*
*&      Form  custom_userexit
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
*      -->C_T_DATA   text
*----------------------------------------------------------------------*
form custom_userexit
          changing c_t_data like t_dp.
  data: lr_zoxcr20022  type zoxcr20022 occurs 0 with header line.
  types: begin of ty_vbap,
           vbeln type vbeln_va,
           posnr type posnr_va,
         end of ty_vbap,
        begin of ty_vbpa,
           vbeln type  vbeln_va,
           posnr type posnr_va,
           parvw type parvw,
           kunnr type kunnr,
           parnr type parnr,
           pernr type pernr_d,
        end of ty_vbpa.

  data: wa_vbap type ty_vbap,
        wa_vbpa type ty_vbpa,
        wa_lines like sy-tabix,
        wa_vbeln type vbeln_va,
        wa_kunnr type kunnr,
        v_posnr type posnr_va,

        ht_vbap type hashed table of ty_vbap with unique key vbeln posnr with header line,
        ht_vbpa type hashed table of ty_vbpa with unique key vbeln posnr parvw with header line,
        ht_vbpa_litems type table of ty_vbpa with header line,
        ht_vbpa_tmp type hashed table of ty_vbpa with unique key vbeln posnr parvw.

  field-symbols:  <fs_c_t_data> type zoxcr20022,
                  <fs_vbpa> type ty_vbpa.

  constants: c_000000 like vbap-posnr value '000000', "Header
             c_ag like vbpa-parvw value 'AG'. "Sold-to ,

  loop at c_t_data assigning <fs_c_t_data>.
    wa_vbap-vbeln = <fs_c_t_data>-vbeln.
* capture header also for subsequent operation
    if wa_vbap-vbeln ne wa_vbeln. "one time for each vbeln
      wa_vbap-posnr = c_000000.
      wa_vbeln = wa_vbap-vbeln.
      insert wa_vbap into table ht_vbap.
    endif.
* end header capture
    wa_vbap-posnr = <fs_c_t_data>-posnr. "capture line
    insert wa_vbap into table ht_vbap.
  endloop.

* sort for all entries selection
  sort ht_vbap by vbeln posnr.
  describe table ht_vbap lines wa_lines.
  clear wa_vbeln.
* process only if there is data
  if wa_lines gt 0.
    select vbeln posnr parvw kunnr parnr pernr
    into corresponding fields of table ht_vbpa
    from vbpa for all entries in ht_vbap
    where vbeln eq ht_vbap-vbeln
      and posnr eq ht_vbap-posnr.

**** separate Header and Line for VBPA
    refresh: ht_vbpa_litems[], ht_vbpa_tmp[] .
    ht_vbpa_litems[] = ht_vbpa[] .
    delete ht_vbpa where posnr <> c_000000 . "Contains header PF
    delete ht_vbpa_litems where posnr =  c_000000 . "Contains lines PF
    sort ht_vbpa by vbeln posnr .
    sort ht_vbpa_litems by vbeln posnr  .
***copying to temp table so we can sort and read Binary on vbeln and PARVW .
    ht_vbpa_tmp[] = ht_vbpa_litems[] .
    free ht_vbap[]."we don’t need this table further

    loop at c_t_data into lr_zoxcr20022.
      move lr_zoxcr20022-posnr to v_posnr .

**Compare Partner at line item  with header partner
** HT_VBPA..if missing then add to item ( lr_ZOXCR20022 )
***if already exists in item... do not overwrite with header.
***example if Header has 10 partners and if Item has 2 of those
**"CHECKED' then add the 2 partners from item and remaining
**8 from the header ...if no partner exists at item level
**then add all the 10 items from the header
      read table ht_vbpa_litems with key  vbeln   = lr_zoxcr20022-vbeln
                                   posnr   = v_posnr
                                   binary search .
      if sy-subrc <> 0 .
****no partners found in item ...need to add to each item from Header to
*** the line item lr_ZOXCR20022
        loop at ht_vbpa where vbeln = lr_zoxcr20022-vbeln
            and posnr = c_000000 .
          lr_zoxcr20022-zzparvw   = ht_vbpa-parvw.
          lr_zoxcr20022-zzkunnr   = ht_vbpa-kunnr.
          lr_zoxcr20022-zzparnr   = ht_vbpa-parnr.
          lr_zoxcr20022-zzpernr   = ht_vbpa-pernr.
          append: lr_zoxcr20022.
        endloop .
      else .
****partner found ... need to add the Partner information from LT_VBPA to the
*** line item lr_ZOXCR20022
        loop at ht_vbpa_litems where vbeln = lr_zoxcr20022-vbeln
            and posnr = v_posnr .
          lr_zoxcr20022-zzparvw   = ht_vbpa_litems-parvw.
          lr_zoxcr20022-zzkunnr   = ht_vbpa_litems-kunnr.
          lr_zoxcr20022-zzparnr   = ht_vbpa_litems-parnr.
          lr_zoxcr20022-zzpernr   = ht_vbpa_litems-pernr.
          append: lr_zoxcr20022.
        endloop.
****Note : there is One Loop in SUBRC <> 0 ... which adds all the header
*** partner to line where NO partner found on the Line item
***in the ELSE part ( ie Subrc = 0 )  we first add the 'checked' partners
*** on the line item then in the loop below we are the REMAINING 'Unchecked'
** from header to line item
        loop at ht_vbpa where vbeln = lr_zoxcr20022-vbeln .

          read table ht_vbpa_tmp transporting no fields
          with table key vbeln = ht_vbpa-vbeln
                         posnr = lr_zoxcr20022-posnr
                         parvw = ht_vbpa-parvw.
          if sy-subrc <> 0 .
            lr_zoxcr20022-zzparvw   = ht_vbpa-parvw.
            lr_zoxcr20022-zzkunnr   = ht_vbpa-kunnr.
            lr_zoxcr20022-zzparnr   = ht_vbpa-parnr.
            lr_zoxcr20022-zzpernr   = ht_vbpa-pernr.
            append: lr_zoxcr20022.
          endif .
        endloop .
      endif .
***in addition to all above we need to send all the header partners as well
* Capture header level partners as well - once for each document
      if wa_vbeln ne lr_zoxcr20022-vbeln.
        loop at ht_vbpa
          where vbeln = lr_zoxcr20022-vbeln .

          lr_zoxcr20022-posnr   = c_000000.
          lr_zoxcr20022-zzparvw = ht_vbpa-parvw.
          lr_zoxcr20022-zzkunnr = ht_vbpa-kunnr.
          lr_zoxcr20022-zzparnr = ht_vbpa-parnr.
          lr_zoxcr20022-zzpernr = ht_vbpa-pernr.
          append: lr_zoxcr20022.
        endloop.

      endif.
      wa_vbeln = lr_zoxcr20022-vbeln.
    endloop.

    clear wa_vbeln.
    loop at lr_zoxcr20022.
      if lr_zoxcr20022-zzkunnr = '' and lr_zoxcr20022-zzparvw ne ''.
        if wa_vbeln ne lr_zoxcr20022-vbeln. "Get the header sold to party
          read table ht_vbpa
          with table key vbeln = lr_zoxcr20022-vbeln
                   posnr = c_000000
                   parvw = c_ag.
          if sy-subrc eq 0.
            lr_zoxcr20022-zzkunnr = wa_kunnr = ht_vbpa-kunnr.
            modify lr_zoxcr20022.
          endif.
        else.
          lr_zoxcr20022-zzkunnr = wa_kunnr.
          modify lr_zoxcr20022.
        endif.
        wa_vbeln = lr_zoxcr20022-vbeln.
      endif.
    endloop.
  endif. "wa_lines outerlevel check

  refresh c_t_data.
  sort lr_zoxcr20022[] by vbeln posnr zzparvw zzpernr.
  delete adjacent duplicates from lr_zoxcr20022[] comparing vbeln posnr zzparvw zzpernr.
  c_t_data[] = lr_zoxcr20022[].

  free:  ht_vbpa[], ht_vbpa_tmp[], lr_zoxcr20022[].

endform.                    "custom_userexit
